rajko

Do sada smo dosta puta spominjali API i API funkcije, a sada ćemo vas preko pet najpopularnijih trikova upotrebe API-ja bolje upoznati sa ovom tematikom. Application Programming Interface (API) predstavlja skup funkcija, koje programeri mogu koristiti za rad sa komponentama, aplikacijama, ili samim operativnim sistemom. Jedan API interfejs sastavljen je od jedne ili više dinamički povezanih biblioteka (Dynamic Linked Library - DLL), koje obezbeđuju specifične funkcionalnosti. DLL fajlovi su biblioteke funkcija koje aplikacije mogu dinamički linkovati i koristiti u run-time-u, za razliku od statičkog linkovanja tokom kompajliranja aplikacije. Pri tome, u isto vreme neograničeni broj aplikacija može pozivati funkciju iz DLL-a. Glavna prednost je to što potrebna funkcija egzistira samo u jednom fajlu na disku, te se data biblioteka samo jednom učitava u RAM memoriju. Osim toga, biblioteke se mogu nadgrađivati i unapređivati nezavisno od aplikacija.Neke stvari je veoma teško izvesti u "čistom" Visual Basic-u, a istovremeno, iste probleme možete mnogo lakše rešiti uz pomoć Windows API-ja, jer vam na raspolaganju stoji gotovo pet stotina funkcija. Pre nego što budete mogli da koristite API funkcije u Visual Basic-u, potrebno je da ih deklarišemo. Instrukcija "Declare" nam omogućava poziv rutina iz DLL fajlova: 

[Public | Private] Declare Sub name Lib "libname" [Alias "aliasname"] [([lista_arg])]

ili 

[Public | Private] Declare Function name Lib "libname" [Alias "aliasname"] [([lista_arg])] [As type]

Sa svakom ovom deklaracijom obogaćujemo naš Visual Basic. Na primer, mnogo informacija o korisnikovom sistemu možemo dobiti uz pomoć funkcije GetWinFlags. Verzija instaliranog Windows-a može se dobiti funkcijom "GetVersion". Funkcija "GetSystemMetrics" reći će nam npr. da li je na sistemu instaliran miš, koja je veličina ekrana, i sl. 

Sve što je potrebno je da posegnemo za ovim funkcijama i da ih deklarišemo. Potražite na disku fajl "Win32API.txt " koji dolazi uz instalaciju Visual Basic-a, a koji sadrži instrukcije "Declare" za većinu funkcija Windows API-ja. 

Još jedna stvar, budite pažljivi kod korišćenja API-ja jer pozivanje funkcija može biti rizično za vaše aplikacije (kao i za korisnike vaših aplikacija). U narednim prilozima prikazaćemo pet vrlo popularnih API funkcija.

Koju funkciju ćemo prvo odraditi? Pa, za početak bićemo malo destruktivni i kreiraćemo rutinu koja će brisati fajlove. Iako Visual Basic već poseduje komandu koja omogućava brisanje fajlova - instrukcija "Kill", njena upotreba u aplikacijama se ne preporučuje, jer je brisanje na ovakav način permanentno. Kao što svi znamo, korisnici često prave greške, a naš cilj je da ih predupredimo u tome. Pa zato, da korisnici ne bi kasnije okrivili programere za sopstvene greške, nećemo koristiti komandu "Kill" već ćemo kreirati sopstvenu rutinu, koju kasnije možemo koristiti u svim aplikacijama. Za ovu svrhu koristićemo API funkciju SHFileOperation, koja se nalazi u biblioteci "shell32". Prvo što treba da uradimo je da deklarišemo funkciju: 

Private Declare Function SHFileOperation Lib "shell32.dll" (ByRef lpFileOp As SHFILEOPSTRUCT) As Long

Kao što vidite funkcija kao argument koristi tip SHFILEOPSTRUCT, a pošto datog tipa nema unutar Visual Basic-a, potrebno je da ga deklarišemo: 

Private Type SHFILEOPSTRUCT

 hwnd As Long

 wFunc As Long

 pFrom As String

 pTo As String

 fFlags As Integer

 fAnyOperationsAborted As Long

 hNameMappings As Long

 lpszProgressTitle As String

End Type

Inače, funkcija SHFileOperation koristi se i za kopiranje, preimenovanje, premeštanje i brisanje fajl objekata. U našem slučaju iskoristićemo je samo za brisanje fajlova. Da bi sam programski kod bio čitljiviji, kreiraćemo nekoliko konstanti: 

Private Const FO_DELETE = &H3

Private Const FOF_ALLOWUNDO = &H40

Posle ovoga ostalo nam je još da kreiramo rutinu za brisanje: 

Public Sub Brisanje(ByVal FileName As String)

Dim FileStruct As SHFILEOPSTRUCT

Dim X As Long

 With FileStruct

  .fFlags = FOF_ALLOWUNDO

  .pFrom = FileName

  .wFunc = FO_DELETE

 End With

 X = SHFileOperation(FileStruct)

End Sub

Prelazimo na drugi primer upotrebe Windows API funkcija, koji je vezan za proces "spuštanja" (Shut Down) operativnog sistema. Funkcija ili rutina koju ćemo kreirati vrlo je zgodna, pa je možete implementirati u gotovo sve vaše programe. Naročito je dobra kada testirate aplikacije ili hardver, kada vam treba brzi Shut Down sistema.Ovaj zadatak će za nas odraditi API funkcija ExitWindowsEx. Kao i prethodna, i ova funkcija ima višestruku ulogu. Uz pomoć nje možete "odlogovati" (log-off) korisnika, restartovati sistem ili izvršiti Shut Down sistema. Kao što već znamo, prvo ide deklaracija: 

Private Declare Function ExitWindowsEx Lib"user32" (ByVal uFlags As Long, ByVal dwReserved _As Long) As Long

Argument uFlags usmerava funkciju ExitWindowsEx na to šta ona treba da odradi. Za potrebe ove rutine kreiraćemo nekoliko konstanti: 

Private Const EWX_LOGOFF = 0

Private Const EWX_SHUTDOWN = 1

Private Const EWX_REBOOT = 2

Private Const EWX_FORCE = 4

Private Const EWX_POWEROFF As Long = 8

Private Const EWX_RESET = EWX_LOGOFF + EWX_FORCE + EWX_REBOOT

I konačno, evo primera rutine sa svim trima navedenim upotrebama ove funkcije: 

Sub LogOff()

 Dim X As Long

 X = ExitWindowsEx(EWX_LOGOFF, 0)

End Sub

Sub Restart()

 Dim X As Long

 X = ExitWindowsEx(EWX_RESET, 0)

End Sub

Sub ShutDown()

 Dim X As Long

 X = ExitWindowsEx(EWX_SHUTDOWN, 0)

End Sub

I to je to. Ovu rutinu možete ugraditi u instalacione procedure vaših aplikacija koje zahtevaju restartovanje sistema.

U četvrtom nastavku serijala o API funkcijama, pozabavićemo se nečim što se naziva "hot-key" (vrući tasteri), a u suštini označava kombinaciju tastera koju treba istovremeno pritisnuti na tastaturi radi pozivanja određene aplikacije ili neke funkcije aplikacije. Na primer, svi znamo da pod Windows operativnim sistemom, prečica "Alt+Tab" omogućava laki i brzi prelaz između aplikacija. Na isti način možemo isprogramirati da se i naša aplikacija odaziva na određenu kombinaciju tastera. Zato ćemo u sledećem primeru nekoj fiktivnoj aplikaciji dodeliti određenu prečicu. 

Za dodelu prečica aplikaciji koristićemo SendMessage API funkciju. Funkcija SendMessage šalje određene sistemske poruke određenom prozoru, ili prozorima, aplikacije. Kao i u prethodnim primerima, krećemo od deklaracije funkcije: 

Private Declare Function SendMessage _

Lib "user32" Alias "SendMessageA" _

(ByVal hwnd As Long, ByVal wMsg As Long, _

ByVal wParam As Long, lParam As Any) 
As Long

Potom ćemo kreirati i sve potrebne konstante: 

Const WM_SETHOTKEY = &H32

Const VK_ALTPAUSE = &H413

I stižemo do finalnog dela, same funkcije za dodelu prečice. U našem primeru, podesićemo da za našu aplikaciju "hot-key" kombinacija bude kombinacija tastera "Alt" i "Pause". Kada ovo odradimo svaki put kada korisnik naše aplikacije pritisne kombinaciju "Alt+Pause", prozor aplikacije će iskočiti u prvi plan. 

Sub PodesiKombinaciju()

Dim iResponse As Long

 iResponse = SendMessage(Me.hwnd, _

 WM_SETHOTKEY, VK_ALTPAUSE, 0)

 If iResponse <> 1 Then MsgBox 
("Alt+Pause nije validna kombinacija!")

 End If

End Sub

Da li ste do sada već zaboravili stari dobri DOS? Najverovatnije je da je odgovor potvrdan, Windows je postao vaša preokupacija. Da li vam je dosadio multitasking? Da li želite da korisnike vaših programa zadržite samo u vašoj aplikaciji? Ponekad je jako dobro "zarobiti" korisnike unutar vaših aplikacija, a ovo je posebno izraženo kada kreirate bezbednosne programe ili dizajnirate kompjuterske igre (kada je vrlo nezgodna stvar pritisak na "Windows" taster). Zbog svega navedenog, da korisnike vaših programa zadržite samo u okviru date aplikacije, moraćete da sprečite korisnike u korišćenju prečica "Ctrl+Alt+Del", "Ctrl+Esc" (što je ekvivalentno tasteru "Win") i prečice "Alt+Tab". Ovo ćemo odraditi uz pomoć API funkcije SystemParametersInfo. 

Funkcija SystemParametersInfo je dizajnirana za korišćenje sa aplikacijama, kao što je "Control Panel", koja omogućava korisnicima da podešavaju radno okruženje. Sa funkcijom SystemParametersInfo možemo preuzeti ili podesiti široki spektar sistemskih parametara, kao što su: fontovi, prikaz prozora, screen saver, podešavanja miša, korisničkih profila i mnogih drugih. 

Kao i kod prethodnih primera, počinjemo od deklaracije funkcije: 

Private Declare Function SystemParametersInfo _

Lib "user32" Alias "SystemParametersInfoA" _

(ByVal uAction As Long, ByVal uParam As Long, _

ByRef lpvParam As Any, ByVal fuWinIni As Long) _

As Long

Konstantama naš programski kod činimo mnogo čitljivijim: 

Private Const SPI_SETSCREENSAVERRUNNING = 97

Način na koji ćemo sprečiti korišćenje kombinacija "Ctrl+Alt+Del", "Ctrl+Esc" i "Alt+Tab", jeste da "ubedimo" operativni sistem da se trenutno procesira screen saver program. A evo kako to izgleda i programski: 

Sub SpreciCtrlAltDelete(bDisabled As Boolean)

Dim lpvParam As Boolean

Dim X As Long

 X = SystemParametersInfo _

 (SPI_SETSCREENSAVERRUNNING, _

 bDisabled, lpvParam, 0)

End Sub

Stigli smo do kraja ovog našeg malog serijala o API funkcijama i njihovoj upotrebi. A u poslednjem nastavku bavićemo se API funkcijom koja nam može pomoći kod potrebe pokretanja drugih aplikacija iz našeg programa. Ponekad će nam zatrebati da pokrenemo neki drugi Windows program, da recimo otvorimo neki dokument, štampamo dokumente, pretražujemo foldere, i slično, a sve to pod Windows operativnim sistemom. Za potrebu pokretanja drugih aplikacija na raspolaganju nam stoji ShellExecute API funkcija. 

Pre nego ovu funkciju upotrebimo u našem programu, potrebno je da je deklarišemo: 

Private Declare Function ShellExecute _

Lib "shell32.dll" Alias "ShellExecuteA" _

(ByVal hWnd As Long, 
ByVal lpOperation As String, _

ByVal lpFile As String, 
ByVal lpParameters As String, _

ByVal lpDirectory As String, 
ByVal nShowCmd As Long) As Long

Const SW_MAXIMIZE = 3

Kada smo ovu funkciju deklarisali, možemo je i pozvati unutar novoformiranog potprograma (IzvrsiProgram): 

Sub IzvrsiProgram(ByVal sFileName as String, _

ByVal sWorkDir as String)

Dim iGreska As Long

Dim sGreskaPoruka As String

 iGreska = ShellExecute(0, "open", _

  sFileName, "", sWorkDir, SW_MAXIMIZE)

 If iGreska <= 32 Then

  Select Case iGreska

   Case 2

    sGreskaPoruka = "Fajl nije pronadjen."

   Case 31

    sGreskaPoruka = "Nema aplikacije 
koja je u vezi " & _

     "sa datom ekstenzijom fajla."

  End Select

  MsgBox (sGreskaPoruka)

 End If

End Sub

Evo jednog vrlo zanimljivog "parčeta" programskog koda, koji vam može pomoći kod kreiranja pomoći, help-a, za krajnje korisnike vaših programa. Radi se o tome da direktno iz vaše aplikacije možete pozvati prethodno pripremljeni HTML help dokument, i to upravo na određenoj (željenoj) temi. Ova procedura takođe može i da zatvori help fajl pri "spuštanju" aplikacije koja je pozvala sam help. 

Za korišćenje procedure, samo pozovite funkciju "PozoviHTMLHelp", i to sa argumentima kao što je putanja do fajla i dodatnim opcijama kao što je Form1.hWnd (što znači ako se forma zatvara, zatvara se i help fajl), plus opciono - broj teme. U zavisnosti uspešnosti, funkcija će vratiti True ili False. 

Evo kako bi izgledao poziv funkcije: 

funk = PozoviHTMLHelp("c:\mojhelp.chm", Form1.hWnd, 124)

A evo i same funkcije: 

Public Const HH_DISPLAY_TOPIC = &H0

Public Const HH_HELP_CONTEXT = &HF

Public Declare Function HtmlHelp Lib "Hhctrl.ocx" _

    Alias "HtmlHelpA" (ByVal hWndCaller As Long, _

    ByVal pszFile As String, ByVal uCommand As Long, _

    ByVal dwData As Long) As Long

Public Function PozoviHTMLHelp(HelpFile As String, _

    Optional WindowHandle As Long, Optional Topic As Long) As Boolean

    Dim lngReturn As Long

    If Len(Dir(HelpFile)) > 0 Then

                If Topic = 0 Then

            lngReturn = HtmlHelp(WindowHandle, HelpFile, HH_DISPLAY_TOPIC, 0)

        Else

            lngReturn = HtmlHelp(WindowHandle, HelpFile, HH_HELP_CONTEXT, Topic)

        End If

               PozoviHTMLHelp = CBool(lngReturn)

       End If

End Function

